La classe de 6ième et le puzzle de Dudeney

Renée De Graeve

19 décembre 2019

0.1 Exercice d'un livre de 6ième

Dans un livre de Mathématiques de la classe de 6ième au chapitre `Angles`, il y a l'exercice suivant intitulé **Le puzzle de Dudeney :**
Soient un triangle équilatéral ABC de côté 10cm et E le point du segment BC tel que EC=5/2cm.
On note M (resp N) le milieu de AB (resp AC).
Faire la figure suivante formée par les 4 polygones colorés :



Montrer que EE_1 et MM_1 sont perpendiculaires à EF, puis découper les 4 morceaux obtenus et assembler ces morceaux pour former un carré : vous avez réalisé le puzzle Dudeney.
Commentaire : ce n'est pas de la géométrie c'est de l'à peu près ! : les valeurs des angles étant des valeurs approchées les pièces du puzzle se chevauchent pour obtenir une figure qui ressemble à un carré !
Pour réaliser le puzzle, il faut faire pivoter la pièce jaune de 180 degrés autour de M pour amener B en A, M_1 en M_3 et F en M_2, faire pivoter la pièce bleue de 180 degrés autour de N, pour amener C en A, E_1 en E_3 et E en E_2, et effectuer une translation de vecteur $\overrightarrow{EE2}$ de la pièce rouge.
Pour que le puzzle soit correct il faut donc que M_2E_2=EF. Puisque M_2A=FB et AE_2=CE=$5/2$ on en déduit que EF=M_2E_2=M_2A+AE_2=$FB+CE$=$BC-EF$ donc EF=$BC/2$ et comme EC=$BC/4$, on en déduit que BF=$BC/4$ et donc le point F est donc la projection de M sur BC.
On pose $h = NE$ et $2a = BC$, on a :tan$(\widehat{EFN}) = NE/FE = h/a = \sqrt{3}/2$.
D'après les valeurs de l'exercice $\widehat{EFN} = 41$ degrés soit $41\pi/180$ radians.
On a : $\widehat{EFN} = \text{atan}(\sqrt{3}/2) \simeq 0.7137243789$ et $41 * \pi/180 \simeq 0.7155849933$
Comme le point F est la projection de M sur BC, $FENM$ est un **rectangle**.
M_1 et E_1 sont symétriques par rapport à I centre de ce rectangle, on a :
MM_1=E_1E et M_1I=IE_1 donc M_1E_3=$2(IE_1 + E_1N)$=$2IN$=$2IM > 2IM_1$.
$M_1M_3E_4E_3$ a 4 angles droits : c'est un rectangle mais ce n'est pas un carré puisque

$M_1M_3 = 2MM_1$ et $M_1E_3 = 2IN = FN = 2IM > 2IM_1$.
On peut aussi montrer que $M_1M_3E_4E_3$ n'est pas un carré avec un calcul d'aires :
l'aire de ABC est : $2ha = a^2\sqrt{3} = 10^2\sqrt{3} \simeq 173.205080757$ cm2.
l'aire du carré de côté FN est : $FN^2 = h^2 + a^2 = 3a^2/4 + a^2 = 175$ cm2.
l'aire du carré de côté $2EE_1$ est : $4a^2\sin(\text{atan}(h/a))^2 \simeq 171.428571429$ cm2.
Remarque Pour faire le puzzle de Dudeney dans le cas général il faut que le quadrilatère $FENM$ soit un parallélogramme vérifiant FN=$2MM_1$=$2EE_1$=$\sqrt{2ha}$ car FN sera le côté du carré. Pour faire cela on se reportera aux sections suivantes.
Conclusion En 6ième, il faut plutôt faire l'exercice avec comme hypothèse : ABC triangle équilatéral, M milieu de AB, N milieu de AC, F projection de M sur BC, E projection de N sur BC, M_1 projection de M sur FN, E_1 projection de E sur FN et faire réaliser le puzzle d'un rectangle.
Ou encore faire le puzzle de Dudeney avec un triangle ABC vérifiant $BC = AH$ lorsque H est le pied de la hauteur issue de A.
En effet, dans ce cas le rectangle $FENM$ est un carré et les points E_1, M_1, I sont confondus : on a alors : $FN = 2MM_1 = 2EE_1$.

0.2 Une activité autour du puzzle de Dudeney

Exercice niveau 4ième ou 3ième
Soit un triangle isocèle ABC de sommet A ($AB = AC$) et soit H le pied de la hauteur issue de A vérifiant $BC = AH = 2a_1$ (par exemple a_1=6 cm).
1/ On veut faire un puzzle permettant de transformer ce triangle en un carré.
On aimerait que ce puzzle ait le plus petit nombre de pièces. Calculer l'aire de ce triangle et en déduire la longueur des côtés que doit avoir le carré.
2/ On ne suppose plus le triangle ABC isocèle mais on suppose toujours que $AH = BC = 2a_1$. Faire bouger le point A sur la parallèle à BC passant par le point de coordonnées $(0, 2a_1)$ et dire quand votre solution reste valable.
Un exemple de solution possible avec 4 pièces
On fait le dessin (on note c_1 l'abscisse de `H`). Si ABC est isocèle on a $c_1 = a_1$.
Avec `Xcas`, on tape :

```
supposons(a1=[5.0,0,10,0.1]);
supposons(c1=[5.0,0,10,0.1]);
B:=point(0,0);C:=point(2*a1);A:=point(c1+i*2*a1);
triangle(A,B,C);
M:=milieu(A,B);N:=milieu(C,A);J:=point(a1+i*a1);
P1:=affichage(polygone(C,B,J),1+rempli):;P1;
P4:=affichage(polygone(A,M,N),4+rempli):;P4;
P2:=affichage(polygone(B,M,J),2+rempli):;P2;
P3:=affichage(polygone(C,N,J),3+rempli):;P3;
F:=projection(droite(C,B),M);E:=projection(droite(C,B),N);
I:=milieu(F,N);affichage(symetrie(I,P4),4+rempli);
affichage(translation(E-M,P2),2+rempli);
affichage(translation(F-N,P3),3+rempli);
A:=A;C:=C;E:=E;M:=M;J:=J;H:=point(c1);A1:=symetrie(I,A);
segment(N,E,affichage=ligne_tiret_point);
segment(M,F,affichage=ligne_tiret_point);
```

```
d:=droite(y=2*a1);
```

On obtient lorsque $c_1 = a_1 = 5$:



Lorsque l'on déplace A sur la droite d, on remarquera que $MNEF$ est un carré et que le point I est à la fois la projection de M sur FN et projection de N sur FN. Pour que le découpage ci-dessus soit valable, il faut que E et F appartiennent au segment BC. E est le milieu de HC et F est le milieu de BH. Cette construction n'est donc valable que si $0 \leq c \leq 2a$.
On obtient lorsque $a_1 = 5$ et $c_1 = 8$:



0.3 Le puzzle de Dudeney de 4 pièces pour ABC isocèle de sommet A et $BC = AH = 2a$

On fait le dessin (on pose $BC = 2a$ et on note c l'abscisse de H pied de la haurteur issue de A). Si ABC est isocèle on a $c = a$.
Avec Xcas, on tape :

```
supposons(a=[5.0,0,10,0.1]);
supposons(c=[5.0,0,10,0.1]);
B:=point(0,0);C:=point(2*a);A:=point(c+i*2*a);
triangle(A,B,C);
M:=milieu(A,B);N:=milieu(C,A);
F:=projection(droite(C,B),M);E:=projection(droite(C,B),N);
I:=milieu(F,N);
P1:=affichage(triangle(F,E,I),1+rempli);
P2:=affichage(polygone(B,F,I,M),2+rempli);
P3:=affichage(polygone(C,N,I,E),3+rempli);
P4:=affichage(polygone(M,I,N,A),4+rempli);
M2:=symetrie(M,I);F2:=symetrie(M,F);
N2:=symetrie(N,I);E2:=symetrie(N,E);
affichage(symetrie(M,P2),2+rempli);
affichage(symetrie(N,P3),3+rempli);
affichage(translation(2*a*i,P1),1+rempli);
d:=droite(y=2*a);
A:=A;M:=M;N:=N;E:=E;F:=F;I:=I;
```

On obtient lorsque $c = a = 5$:



Comme précédemment, pour que le découpage ci-dessus soit valable, il faut que E et F appartiennent au segment BC. E est le milieu de HC et F est le milieu de BH. Cette construction n'est donc valable que si $0 \leq c \leq 2a$.

On obtient lorsque $a = 5$ et $c = 8$:



0.4 La solution de Dudeney pour un triangle ABC $BC = 10$ $AH = 5\sqrt{3}$

Soient le triangle ABC verifiant $BC = 10$ $AH = 5\sqrt{3}$, M (resp(N) le milieu de AB (resp AC et F_0 (resp E_0) la projection de M (resp N) sur BC.



On a vu précédemment que si F_0E_0NM est un rectangle alors le puzzle de Dudeney transforme ABC en un carré.
Soit I le centre du rectangle F_0E_0NM. M_1 et E_1 sont symétriques par rapport à I et $MM_1 = EE_1 < MI = IN$. Puisque $EN = FM < FE = MN$, le point M_1 se trouve entre F et I et le point E_1 se trouve entre I et N.
Soit $\overrightarrow{u}$ le vecteur unitaire de $\overrightarrow{BC}$.
On translate F_0 en F et E_0 en E par une translation de vecteur $k\overrightarrow{u}$, le rectangle $FENM$ se transforme en un parallélogramme (voir la figure ci-dessus).
Il faut bien sûr que $-k \leq c/2$ et $c/2 + k + a \leq 2a$ pour que F et E restent sur le segment BC.
Pour que $M_1, E_2E_4M_2$ soit un carré, il suffit de choisir k pour que $FN = 2MM_1$.
Si on choisit $k < 0$, F se rapproche de B, la longueur de FN augmente alors que celle de EM diminue : on a alors $MM_1 + EE_1 < ME < FN$.
Donc $k >= 0$.
On pose $BC = 2a$, $AH = 2h$, $BH = c$ avec $0 \leq c \leq 2a$.
Si le triangle ABC est équilatéral, on a : $h = a\sqrt{3}/2$ et $c = a$ mais on fait le dessin de façon à voir comment évolue la figure lorsque on ne suppose plus le triangle ABC équilatèral mais lorsque on suppose seulement que $BC = 2a$ et $AH = a\sqrt{3}/2$.
Donc on doit avoir $-k \leq c/2$ et $c/2 + k + a \leq 2a$.
On rajoute donc les paramètres c et k ($BH = c$ et $BF = c/2 + k$) pour pouvoir faire bouger le point A sur la parallèle à BC passant par le point de coordonnées $(0, a\sqrt{3})$).
On tape avec `Xcas` :

```
supposons(c=[5,0,10,0.1]);;
supposons(k=[0.06,(-5)/2,5/2,0.01]);;
B:=point(0);;
C:=point(10);;
triangle_equilateral(B,C,A);;
M:=milieu(B,A);;
N:=milieu(C,A);;
F:=point(5/2+k);E:=point(15/2+k);;
d1:=droite(F,N):;d1:=d1;;
E1:=projection(d1,E);;
M1:=projection(d1,M);;
P1:=polygone(F,E,E1):;P1;;
P3:=polygone(N,E1,E,C):;P3 ;;
P2:=polygone(B,F,M1,M):;P2;;
P4:=polygone(N,A,M,M1):;P4;;
M2:=symetrie(M,M1);;
E2:=symetrie(N,E1);;
E3:=symetrie(N,E);;
F3:=symetrie(M,F);;
E4:=translation(E3-E,E1);;
symetrie(N,P3);;
symetrie(M,P2);;
translation(E3-E,P1);;
segment(N,15/2,affichage=ligne_tiret);;
segment(M,F),segment(N,E),segment(M,N);;
segment(M,5/2,affichage=ligne_tiret);;
I:=milieu(F,N,affichage=quadrant2);
```

On obtient pour $AB = BC = 10 = 2a$, $AH = a\sqrt(3) = 2h =$, $BH = c = 5$ et $k = 0.045$:



Calculons k en fonction de a et h
On a : $FN^2 = (a-k)^2 + h^2$
L'aire du triangle FEE_1 vaut $FN * EE_1/2 = h * FE/2 = ha/2$ donc
$FN * EE_1 = ha$.
On veut avoir $EE_1 = FN/2$, soit :
$FN * 2EE_1 = FN^2 = 2ha = (a-k)^2 + h^2$ pusque FNE_0 est rectangle en E_0.
k doit donc vérifier : $(a-k)^2 = 2ha - h^2 = h(2a-h)$ ce qui impose $h \leq 2a$.
On a $a^2 - 2ha + h^2 = (a-h)^2 \geq 0$ et $a > 0$ donc $a \geq \sqrt{2ha - h^2}$
On a donc bien $k \geq 0$.
Si $h \leq 2a$ on a $k = a - \sqrt{2ha - h^2}$
Puisque $E \in [B, C]$ on doit avoir $-k < c/2+a+k \leq 2a$ Pour que M_1 et E_1 soient sur le segment FN il suffit que E_1 soit sur le segment FN puisque M_1 et E_1 sont symétriques par rapport à I centre du parallélogramme. EE_1 est perpendiculaire à FN donc E_1 se trouve sur le cercle de diamètre FE. N a pour ordonnée h et $h > a/2$ donc le cercle de diamètre FE se trouve en dessous de la droite $y = h$ donc l'intersection de FN et du cercle se trouve entre F et N.
On peut calculer les coordonnées de E_1, on a :
A (resp B et C) est le point d'affixe $c + 2ih$ (resp 0 et $2a$),
F_0 (resp E_0) est le point d'affixe $c/2$ (resp $a + c/2$),
F (resp E) est le point d'affixe $c/2 + k$ (resp $a + c/2 + k$),
M (resp N) est le point d'affixe $c/2 + ih$ (resp $a + c/2 + ih$),
Soit P la projection de E_1 sur FE (E_1 est sur le cercle de diamètre FE donc P se trouve entre F et E et on a $FP = FE - EP$,
les triangles FE_0N et E_1PE sont semblables.
Puisque $FN = 2EE_1$, donc on a : $E_0N = 2PE = h$ et $FE_0 = 2E_1P = a - k$
donc $E_1P = (a-k)/2 = \sqrt{2a * h - h^2}/2$, $PE = h/2$,
$BP = BF + FP = BF + FE - EP = c/2 + k + a - h/2$ donc :
E_1 a pour coordonées : $c/2 + a + k - h/2$, $\sqrt{2a * h - h^2}/2$.
Pour que E_1 soit entre F et N il faut que $\sqrt{2a * h - h^2}/2 \leq h$ i.e $2a*h-h^2 \leq 4h^2$
ce qui entraine $2a/5 \leq h$.
Donc on doit avoir $2a/5 \leq h \leq 2a$.
On tape :
`a:=5;h:=a*sqrt(3)/2;k:=a-sqrt(2h*a-h^2)))`
On obtient :
`5,5*sqrt(3)/2,5-(sqrt(25*(-3+4*sqrt(3))/4))`
`5-(sqrt(25*(-3+4*sqrt(3))/4))` $\simeq$ `0.0450761671624`.
Pour voir les pièces du puzzle, on remplace :
`supposons(k=[0.19,(-5)/2,5/2,0.01])` par
`k:=5-sqrt(25*sqrt(3)-75/4)`
et on tape :

```
affichage(P1,1+rempli)
translation(F3-F,P1),1+rempli)
affichage(P2,2+rempli)
affichage(symetrie(M,P2),2+rempli)
affichage(P3,3+rempli)
affichage(symetrie(N,P3),3+rempli)
affichage(P4,4+rempli)
```

Pour faire bouger le point A sur `d:= droite(y=5*sqrt(3))`, on remplace `triangle_equilateral(B,C,A);` par `A:=point(c,5*sqrt(3))`.
Pour $c = 7.7$ on obtient :



0.5 La solution de Dudeney en général

On considère un triangle quelconque ABC. On pose $BC = 2a$, $AH = 2h$, $BH = c$, $k = a - \sqrt{2ha - h^2}$
Voici la figure pour $a = 6$, $h = 8$, $c = 9$:



Avec le système de coordonnées ayant pour origine B et pour axe des x $\overrightarrow{BC}$, on a :
`B:=point(0);C:=point(2*a);A:=point(c,2h);H:=point(c);F:=point(c/2+k);`
Cherchons les coordonnées de M_1 : 1/ géométriquement : soient F_0 et E_0 les projections de M et N sur BC.

Les triangles rectangles M_1MH0 et E_0FN sont semblables car $\widehat{H_0MM_1} = \widehat{NFC}$ (angles à côtés perpendiculaires) si on pose $MH_0 = h_1$, on a $FN = 2MM_1$ donc :

$h_1/FE_0 = H_0M_1/NE_0 = MM_1/FN = 1/2$ donc :
$h_1 = FE_0/2 = (a-k)/2$ et $H_0M_1 = NE_0/2 = h/2$ donc :
M_1 a pour coordonnées $c/2+h/2, h-h_1 = h-(a-k)/2 = h-\sqrt{2ah-h^2}/2$
2/ on demande à `Xcas` de calculer l'affixe de M_1, on tape :
`evalc(simplify(affixe(M1)))`
On obtient :
`h/2+c/2+i*(-(sqrt(-h^2+2*a*h))+2*h)/2`

Les conditions que doivent vérifier a, h, c pour que le puzzle soit possible :

1. pour que k soit défini il faut que $h \leq 2a$
2. Les points E et F doivent être sur le segment BC, donc on doit avoir :
$$c/2+k \geq 0 \text{ et } c/2+a+k \leq 2a$$
3. Les points M_1 et E_1 doivent être sur le segment FN. Puisque M_1 et E_1 sont symétriques par rapport au centre du parallélogramme $NMFE$, si M_1 est sur le segment FN, E_1 l'est aussi donc on doit avoir :
$0 \leq h_1 = h-\sqrt{2ah-h^2}/2 \leq h$ i.e. $\sqrt{2ah-h^2} \leq 2h$ donc
$2ah-h^2 \leq 4h^2$ i.e. $2ah \leq 5h^2$ donc la condition est :
$$2a/5 \leq h \leq 2a$$
4. Les points F et E doivent être sur le segment BC si $\overrightarrow{B_1B} = k$, donc on doit avoir :
$-k \leq c/2$ et $c/2+a+k \leq 2a$ i.e.
$$-2k \leq c \leq 2\sqrt{2ah-h^2}$$

Avec **Xcas**, on tape :

```
supposons(a=[5.9,0,10,0.1]);
supposons(h=[7.9,2*a/5,2*a,0.1]);
k:=a-sqrt(2*a*h-h^2);
supposons(c=[8.8,-2*k,2*sqrt(2*a*h-h^2),0.1]);
A:=point(c+2*h*i,affichage=quadrant2);
B:=point(0,0,affichage=quadrant1);
C:=point(2*a,affichage=quadrant1);
M:=milieu(A,B);
N:=milieu(A,C);
k:=a-sqrt(2*a*h-h^2);
F:=point(c/2+k);
E:=point(c/2+a+k);
triangle(A,B,C);
s:=segment(F,N);
M1:=projection(s,M);
E1:=projection(s,E);
segment(M,M1);
segment(E,E1);
M2:=symetrie(M,M1);
E2:=symetrie(N,E1);
E3:=symetrie(N,E);
```

```
E4:=translation(F1-F,E1);
F1:=symetrie(M,F);
polygone(E4,F1,M2,M1,E2,E3);
segment(M,N);segment(F1,E3);
F0:=affichage(projection(segment(B,C),M),1+quadrant2);
E0:=affichage(projection(segment(B,C),N),2+quadrant2);
H0:=affichage(projection(segment(F0,M),M1),1);
segment(F0,M,affichage=1+);
triangle(M,M1,H0,affichage=1);
 triangle(F,E0,N,affichage=2);
polygone(E1,F,E,affichage=1+rempli);
polygone(E4,F1,E3,affichage=1+rempli);
angle(F,E,N,"",affichage=2);
 angle(M,H0,M1,"",affichage=1);
polygone(B,F,M1,M,affichage=2+rempli);
polygone(A,F1,M2,M,affichage=2+rempli);
polygone(C,N,E1,E,affichage=3+rempli);
polygone(A,N,E2,E3,affichage=3+rempli);
polygone(A,M,M1,N,affichage=4+rempli);
M:=M;N:=N;A:=A;M1:=M1;
E1:=E1;F1:=F1;M2:=M2;
```





0.6 Comment faire le dessin avec une règle et un compas

Pour transformer un triangle de base $2a$ et de hauteur $2h$ en un carré il est facile de calculer la longueur l du carré en égalant l'aire du triangle à l'aire du carré : $S = 2ha = l^2$ donc $l = \sqrt{2ha}$.
Ce calcul présente un inconvénient car la valeur de l sera approchée.
On peut avoir une procédure purement géométrique :
Soit Q le point(0) et QR un segment s longueur $2a + h$ et un point H de s tel que : $QH = 2a$ et $HR = h$.
Voilà la figure pour $a = 6$ et $h = 8$:



On construit le point P sur d la demi-droite HK directement perpendiculaire à QR passant par H .
On veut que le triangle PQR soit rectangle en P donc P est l'intersection de d et du cercles de diamètre QR.
Les triangles QHP et PHR sont semblables, donc $PH^2 = HQ * HR = 2ah$ et $PH = l$.
Le point F du puzzle de Dudeney est donc l'intersection du cercle de centre N et de rayon l avec le segment BC.
On peut ensuite construire aisement le point E puisque $FE = a$.
La valeur de k sera représentée par $F_0F = a - l$.

0.7 Une animation

On tape :

```
animtri(t1,t2,t3):={
local L,l,M,P,P1,Q,R;
l:=sqrt(4*sqrt(3.)/3-1);
M:=point(1):;P:=point(i*(2-l)):;P1:=symetrie(M,P):;
Q:=point(2+i*l):;R:=rotation(M,pi/3,Q):;
L:=quadrilatere(M,2,Q,R,affichage=4+rempli);
L:=L,affichage(rotation(M,t1,quadrilatere(0,M,R,i*(2-l))),
              1+rempli);
L:=L,affichage(rotation(Q,-t3,quadrilatere(Q,2+2*i,1+2i,R)),
              3+rempli);
L:=L,affichage(translation((P1-P)*t2,triangle(i*(2-l),2i,1+2i)),
              2+rempli);
return L;
}:;
T1:=seq([animtri(t1,0,0)],t1=0..3.14,3.14/10):;
```

```
T3:=seq([animtri(3.14,0,t3)],t3=0..3.14,3.14/10):;
T2:=seq([animtri(3.14,t2,3.14)],t2=0..1.0,0.1):;
T4:=seq([animtri(t1,1,3.14)],t1=3.14..0,3.14/10):;
T6:=seq([animtri(0,1,t3)],t3=3.14..0,3.14/10):;
T5:=seq([animtri(0,t2,0)],t2=1..0,0.1):;
```

Puis, on tape : `animation(T1,T3,T2,T4,T6,T5)`
On obtient la transformation d'un carré en un triangle équilatéral.

0.8 Le pavage fait avec les 4 pièces

Voici le pavage réalisé avec les 4 pièces du puzzle de Dudeney pour le triangle ABC avec `A:=point(0);B:=point(10);C:=point(2,7)`.



0.9 Trois puzzles différents pour un même triangle

Soit le triangle ABC et H le pied de la hauteur issue de A vérifiant :
$BC = 2a = 10$, $AH = 2h = 4$ et $BH = 3$.
Façon 1 : on fait le puzzle pour le triangle A, BC avec $BC = 2a = 10$, $AH = 2h = 4$ et $BH = c = 3$ et on obtient le puzzle 1.
Façon 2 : on peut faire aussi le puzzle pour le triangle C, AB, on a si H_1 est le pied de la hauteur issue de C, on a :
$AB = 2a_1$, $CH_1 = 2h_1$ et $AH_1 = c_1$ et on obtient le puzzle 2.
Façon 3 : on peut faire aussi le puzzle pour le triangle B, CA, on a si H_2 est le pied de la hauteur issue de B, on a :
$CA = 2a_2$, BH_2 et $AH_2 = c_2$ et on obtient le puzzle 3.

Puzzle 1 :



Puzzle 2 :



Puzzle 3 :



0.10 Un Exercice

Soit le triangle ABC et H le pied de la hauteur issue de A vérifiant :
$AB = 10 = 2a$, $AH = 3 = 2h$ et $BH = 5/2 = c$.
Faire le puzzle de Dudeney de 2 façons différentes : on ne peut pas faire le Puzzle1 car $h = 3/2 < 2a/5 = 2$.
On obtient :
Puzzle 2 pour le triangle C, AB :



Puzzle 3 pour le triangle B, CA :



0.11 Les puzzles faciles à réaliser

On peut choisir d'avoir un entier comme valeur de k par exemple (car on a $3^3 + 4^2 = 5^2$ et $5^2 + 12^2 = 13^2$) :
$a = 5$, $h = 2$, on a alors $k = 5 - \sqrt{20 - 4} = 5 - 4 = 1$
$a = 5$, $h = 8$, on a alors $k = 5 - \sqrt{80 - 64} = 5 - 4 = 1$
$a = 5$, $h = 9$, on a alors $k = 5 - \sqrt{90 - 81} = 5 - 3 = 2$
$a = 13$, $h = 8$, on a alors $k = 13 - \sqrt{16 * 13 - 64} = 13 - 12 = 1$
$a = 13$, $h = 18$, on a alors $k = 13 - \sqrt{36 * 13 - 18^2} = 13 - 12 = 1$
$a = 13$, $h = 25$, on a alors $k = 13 - \sqrt{50 * 13 - 25^2} = 13 - 5 = 8$